Доброго времени суток всем читающим! Друзья подскажите как вывести макет скд со своим запросом в табличное поле внешней обработки. Макет своял, проверил в консоли отчетов выведение нормальное, но вот почему то найти как прикрутить к табличному полю этот макет не получается, подскажите, что еще не хватает тут?

```1c
    СхемаКомпоновкиДанных = ЭтотОбъект.ПолучитьМакет("Макет1");
    КомпоновщикНастроек = Новый КомпоновщикНастроекКомпоновкиДанных;
    КомпоновщикНастроек.Инициализировать(Новый ИсточникДоступныхНастроекКомпоновкиДанных(СхемаКомпоновкиДанных));

    ИсполняемыеНастройки = КомпоновщикНастроек.Настройки;
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("номенклатура", ЭлементыФормы.Список.ТекущаяСтрока);
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("Период", ТекущаяДата());
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("НаДату", ТекущаяДата());
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("СП", СП);

    ТабДок = ЭлементыФормы.ОстаткиЦены;
    ПроцессорВывода=Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(ТабДок);
```
---
Схему компоновки еще бы посмотреть, мне кажется там не настроена структура
---
PhoenixAOD,

```1c
    СхемаКомпоновкиДанных = ЭтотОбъект.ПолучитьМакет("Макет1");

    КомпоновщикНастроекСКД = Новый КомпоновщикНастроекКомпоновкиДанных;
    КомпоновщикНастроекСКД.Инициализировать(Новый ИсточникДоступныхНастроекКомпоновкиДанных(СхемаКомпоновкиДанных));
    КомпоновщикНастроекСКД.ЗагрузитьНастройки(СхемаКомпоновкиДанных.НастройкиПоУмолчанию);

    ТиповыеОтчеты.УстановитьПараметр(КомпоновщикНастроекСКД, "номенклатура", ЭлементыФормы.Список.ТекущаяСтрока);
    ТиповыеОтчеты.УстановитьПараметр(КомпоновщикНастроекСКД, "Период", ТекущаяДата());
    ТиповыеОтчеты.УстановитьПараметр(КомпоновщикНастроекСКД, "НаДату", ТекущаяДата());
    ТиповыеОтчеты.УстановитьПараметр(КомпоновщикНастроекСКД, "СП", СП);

    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;
    МакетКомпоновки = КомпоновщикМакета.Выполнить(СхемаКомпоновкиДанных, КомпоновщикНастроекСКД.Настройки);

    ПроцессорКомпоновкиДанных = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновкиДанных.Инициализировать(МакетКомпоновки);

    ТабДок = ЭлементыФормы.ОстаткиЦены;
    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(ТабДок);

    ПроцессорВывода.Вывести(ПроцессорКомпоновкиДанных);
```
---
shurikvz,
```1c
    КомпоновщикНастроекСКД.ЗагрузитьНастройки(СхемаКомпоновкиДанных.НастройкиПоУмолчанию);
```
 вот этого и не хватало))) Спасибо))))
---
```1c
    СхемаКомпоновкиДанных = ЭтотОбъект.ПолучитьМакет("Макет1");
    КомпоновщикНастроек = Новый КомпоновщикНастроекКомпоновкиДанных;
    КомпоновщикНастроек.Инициализировать(Новый ИсточникДоступныхНастроекКомпоновкиДанных(СхемаКомпоновкиДанных));

    ИсполняемыеНастройки = КомпоновщикНастроек.Настройки;
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("номенклатура", ЭлементыФормы.Список.ТекущаяСтрока);
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("Период", ТекущаяДата());
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("НаДату", ТекущаяДата());
    ИсполняемыеНастройки.ПараметрыДанных.УстановитьЗначениеПараметра("СП", СП);

    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;
    МакетКомпоновки = КомпоновщикМакета.Выполнить(СхемаКомпоновкиДанных, ИсполняемыеНастройки);

    ПроцессорКомпоновки = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновки.Инициализировать(МакетКомпоновки);

    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(ТабДок);
    ПроцессорВывода.Вывести(ПроцессорКомпоновки);
```

Так попробуй
---
ИльяЕвгеньевич, да макет подключился, вывел только параметры и все. просто какая то неведомая фигня

запрос в консоли запросов выводит данные, если перегонять через макет табличный документ тоже все выводит.
---
Спасибо сегодня попробую!
---
Тема все еще актуальна
---
Схему компоновки еще бы посмотреть, мне кажется там не настроена структура
---
Всем спасибо кто откликнулся)